<commit_before>{ lib, rustPlatform, fetchFromGitHub, pkgs, stdenv, Security }:

rustPlatform.buildRustPackage rec {
  pname = "bindle";
  version = "0.8.1";

  src = fetchFromGitHub {
    owner = "deislabs";
    repo = pname;
    rev = "v${version}";
    sha256 = "sha256-Mc3LaEOWx8cN7g0r8CtWkGZ746gAXTaFmAZhEIkbWgM=";
  };

  doCheck = false; # Tests require a network

  buildInputs = with pkgs; [ pkg-config openssl_3 ] ++ lib.optional stdenv.isDarwin Security;

  cargoSha256 = "sha256-brsemnw/9YEsA2FEIdYGmQMdlIoT1ZEMjvOpF44gcRE=";

  cargoBuildFlags = [
    "--bin" "bindle"
    "--bin" "bindle-server"
    "--all-features"
  ];

  meta = with lib; {
    description = "Bindle: Aggregate Object Storage";
    homepage = "https://github.com/deislabs/bindle";
    license = licenses.asl20;
    maintainers = with maintainers; [ endocrimes ];
    platforms = platforms.unix;
  };
}
<commit_msg>bindle: fix build failure<commit_after>{ lib, rustPlatform, fetchFromGitHub, stdenv, Security, pkg-config, openssl }:

rustPlatform.buildRustPackage rec {
  pname = "bindle";
  version = "0.8.1";

  src = fetchFromGitHub {
    owner = "deislabs";
    repo = pname;
    rev = "v${version}";
    sha256 = "sha256-Mc3LaEOWx8cN7g0r8CtWkGZ746gAXTaFmAZhEIkbWgM=";
  };

  doCheck = false; # Tests require a network

  nativeBuildInputs = [ pkg-config ];
  buildInputs = [ openssl ] ++ lib.optional stdenv.isDarwin Security;

  cargoSha256 = "sha256-brsemnw/9YEsA2FEIdYGmQMdlIoT1ZEMjvOpF44gcRE=";

  cargoBuildFlags = [
    "--bin" "bindle"
    "--bin" "bindle-server"
    "--all-features"
  ];

  meta = with lib; {
    description = "Bindle: Aggregate Object Storage";
    homepage = "https://github.com/deislabs/bindle";
    license = licenses.asl20;
    maintainers = with maintainers; [ endocrimes ];
    platforms = platforms.unix;
  };
}

{'fraction_non_alphanumeric': 0.1340599455040872, 'fraction_numerical': 0.0332425068119891, 'mean_word_length': 3.5671641791044775, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4868941', 'n_tokens_mistral': 753, 'n_tokens_neox': 708, 'n_words': 129}